
Abstract

V článku jsou rozvedeny matematické základy modelu sdílení tepla a ochlazování kon-
tejneru s vyhořelým jaderným palivem. Z rozboru modelu je navržena metodika měření
rozhodujících fyzikálních veličin a následně způsob výpočtu zbytkového výkonu.

Priciples of the mathemacial modelling of the heat exchange and cooling of the contai-
ner with spent fission fuel are explained in the presented article. Methodics of the measure-
ment of the main physical quantities and alghoritm of the computation of the remamining
power are based on this model.

Úvod

Palivo po vyjmutí z reaktoru není zcela beze zbytku vyhořelé, tzn. že v něm
stále ještě probíhají jaderné reakce. Důsledkem těchto reakcí je mimo jiné i vznik
tepelné energie, která ohřivá celý palivový článek. Pokud by se článek neochla-
zoval, zvyšovala by se jeho teplota a v případě, že by přesáhla konstrukční limity,
došlo by k jeho destrukci. Znalost zbykového výkonu je proto důležitá pro bez-

pečné skladování vyhořelého jaderného paliva.
Pomocí programu Maple se sestaví matematický model popisující časovou

závislost teploty vnější stěny kontejneru na zbytkovém výkonu paliva, způsobu
ochlazování a okolních podmínkách. Tento model bude možné použít i inverzně
– ze známé teploty kontejneru, způsobu ochlazování a okolních podmínek lze
určit zbytkový výkon paliva.

Vstupní příkazy a odpovědi programu Maple budou sázeny následujícím způ-
sobem

> e1 := Int(x, x=a..b) = int(x, x=a..b);

e1 :=
∫ b

a
x dx = b2 − a2

Matematický model

Pro kontejner s vyhořelým jaderným palivem je možné zapsat zákon zacho-
vání energie, viz např. [1], ve tvaru:

> restart;
> e1 := P*t = K*(T(t)-Ti) + Int(S*alpha*(T(t)-Te),t);

e1 := P t = K (T (t) − Ti) +
∫ t

0
S α(T (t)) (T (t) − Te) dt ,

kde
P = zbytkový výkon T = okamžitá teplota povrchu

t = čas S = plocha povrchu kontejneru

K = tepelná kapacita
kontejneru s palivem

α = koeficient přestupu tepla,
obecně závisí na teplotě
povrchu

Ti = počáteční teplota povrchu Te = teplota okolí

Integrál z rovnice e1 je možné odstranit derivací této rovnice podle času a z vý-
sledku vyjádřit derivaci teploty podle času d T(t)/d t

> e2:=diff(e1,t);

e2 := P = K
d

dt
T (t) + S α(T (t)) (T (t) − Te)

> e3:=diff(T(t),t)=solve(e2,diff(T(t),t));

e3 :=
d

dt
T (t) = −S α(T (t)) (T (t) − Te) − P

K
.

Rovnice e3 je obyčejnou diferenciální rovnicí pro časovou závislost teploty,
kterou ovšem výrazně komplikuje obecně neznámá závislost koeficientu pře-
stupu tepla na teplotě α(T (t)). Nicméně pomocí programu Maple je možné nalézt
její řešení.

> dsol1:=dsolve(e3,T(t));

dsol1 := t −
∫ T (t) K

−Sα (_a) _a + Sα (_a)Te + P
d_a + _C1 = 0 . (1)

Maple nalezl řešení rovnice e3 ve tvaru inverzní funkce t (T ), navíc je nutné
znát funkční závislost α(T (t)), bez kterého nelze v analytickém řešení rovnice (1)
dále pokračovat.

Vrat’me se proto k rovnici e3. Vzhledem k tomu, že koeficient přestupu tepla
α(T (t)), je neznámou funkcí teploty, ale teplota samotná je funkcí času, její funkční
závislost také samozřejmě neznáme, je možné použít substituce α(T (t)) = β(t), čili
konstatovat, že koeficient přestupu tepla je neznámou fukcí času. Rovnice e3 tak
přejde na tvar jednodušší diferenciální rovnice, pro kterou již lze nalézt analytické
řešení.

> e4 := subs(alpha(T(t))=beta(t),e3);

e4 :=
d

dt
T (t) = −S β(t) (T (t) − Te) − P

K

> dsol2 := simplify(dsolve(e4,T(t)));

dsol2 := T (t) =

∫
e

S
K

∫
β(t) dt (S β(t) Te + P ) dt + _C1 K

K e
S
K

∫
β(t) dt

. (2)

Stále zůstává problémem neznalost závislosti koeficientu přestupu tepla na
čase, nicméně rovnice (2) již obsahuje přímý vztah pro T (t). V dalším kroku se
pokusíme určit integrační konstantu _C1.

Nejprve je nutné doplnit ve všech integrálech integrační meze. Dále je možné
předpokládat, že v průběhu měření se teplota stěny kontejneru bude pohybovat
pouze v úzkem rozmezí, díky tomu je možné nahradit neznámou funkci závis-
losti koeficientu přestupu tepla na čase, nebo na teplotě, střední hodnotou

a =

∫ t
0 β (t) dt

t
=

∫ Tf
Ti α (T ) dT

Tf − Ti
.

> T1 := T(t) = (Int(exp(1/K*S*Int(beta(t),t=0..t))*
> (S*beta(t)*Te+P),t=0..t)+_C1*K)*
> exp(-1/K*S*Int(beta(t),t=0..t))/K:
> T2 := subs(beta(t)=a, T1);

T2 := T (t) =

∫ t

0
e

S
K

∫ t
0 a dt (S a Te + P ) dt + _C1 K

K e
S
K

∫ t
0 a dt

. (3)

Rovnici (3) je možné vyjádřit v analytickém tvaru

T3:=collect(expand(value(T2)),exp);

3



T3 := T (t) = Te +
P

a S
+

−Te − P

a S
+ _C1

e
a S t
K

(4)

Integrační konstantu _C1 určíme z podmínky, limt→0 T (t) = Ti , teplota stěny
kontejneru v čase t = 0 je rovna počáteční teplotě Ti. Jednoduchý výpočet vede
k výsledku _C1 = Ti .

Podobným výpočtem pro finalní ustálenou teplotu stěny kontejneru obdržíme
rovnici pro konečnou teplotu stěny kontejneru

> Tf = expand(limit(rhs(T3),t=infinity))
> assuming S>0,K>0,a>0;

Tf = Te +
P

S a
.

Protože ustálenou teplotu stěny kontejneru Ti i teplotu okolí Te lze přesně změřit,
je možné při známém topném výkonu P , který je možné provést například elektric-
kým ohřevem, určit součin a S.

> map(u->(u-Te)*S*a/(Tf-Te),%);

a S =
P

Tf − Te
.

Rovnici (4) je tak možné pomocí substituce

> su := _C1=Ti,P=C1*S*a,K=tau*S*a;

su := _C1 = Ti , P = C1 S a, K = τ S a

přepsat do finálního tvaru

> T = subs(su,rhs(T3));

T = Te + C1 + (Ti − C1 − Te) e
− t

τ . (5)

Určení zbytkového výkonu paliva

V případě, že naměříme teplotu stěny kontejneru v závislosti na čase a známe
je možné pomocí regresní analýzy z rovnice (5) určit neznámé koeficienty C1 a τ .
K výpočtu použijeme metodu nejmenších čtverců.

Předpokládejme, že v čase ti jme naměřili teplotu povrchu kontejneru Yi. Cel-
kový počet naměřených hodnot je N . Nejprve stanovíme druhou mocninu roz-
dílu dTi mezi funkční hodnotou teploty T(ti) a odpovídající naměřenou teplotou
Yi. Pro zvýšení přehlednosti výpočtu použijeme substituce, které zjednoduší něj-
častěji používané výrazy.

> dT[i] := subs(su1,su2,t=t[i],T)-Y[i]:
> dT2[i] := expand(dT[i]^2);

dt2i := C12
(
e

ti
τ

)−2
− 2 C12

(
e

ti
τ

)−1
− 4 C1 Te

(
e

ti
τ

)−1
+ 2 C1 Yi

(
e

ti
τ

)−1

+Ti2
(
e

ti
τ

)−2
+ Te2

(
e

ti
τ

)−2
− 2 Te2

(
e

ti
τ

)−1
+ C12 + 2 C1 Te − 2 C1 Yi

+Te2 − 2 Te Yi + Yi
2 + 2 Te Yi

(
e

ti
τ

)−1
− 2 Ti C1

(
e

ti
τ

)−2
− 2 Ti Te

(
e

ti
τ

)−2

+2 Ti C1
(
e

ti
τ

)−1
+ 2 Ti Te

(
e

ti
τ

)−1
− 2 Ti Yi

(
e

ti
τ

)−1
+ 2 C1 Te

(
e

ti
τ

)−2
.

Proměnná dt2i je dosti nepřehledná, proto ve výpočtu sumy kvadrátu residuí,
SQR, provedeme několik substitucí, které zvýší přehlednost výsledku.

> SQR : =map(u->simplify(sum(u,i=1..N)),dT2[i]):
> sigma := selectfun(SQR,sum):
> Su := [seq(sigma[j]=Sigma[j],j=1..nops(sigma))];
> Bs := map(u->rhs(u)=lhs(u),Su):
> SQR := subs(Su,SQR);

Su :=

[
N∑

i=1

e−
ti
τ Yi = Σ1,

N∑
i=1

Yi = Σ2,
N∑

i=1

e−
ti
τ = Σ3,

N∑
i=1

Yi
2 = Σ4,

N∑
i=1

e−2
ti
τ = Σ5

]

SQR := C12Σ5 − 2 C12Σ3 − 4 C1 Te Σ3 + 2 C1 Σ1 + Ti2Σ5 + Te2Σ5 − 2 Te2Σ3

+NC12 + 2 NC1 Te − 2 C1 Σ2 + NTe2 − 2 Te Σ2 + Σ4 + 2 Te Σ1

−2 Ti C1 Σ5 − 2 Te T i Σ5 + 2 Ti C1 Σ3 + 2 Te T i Σ3 − 2 Ti Σ1 + 2 C1 Te Σ5 .

Nyní je nutné nalézt takovou hodnotu koeficientu C1, která minimalizuje
sumu kvadrátu residuí SQR. Z této hodnoty je pak možné určit odpovídající vý-
kon, případně provést zpětnou substituci, jejíž výstup zde není uveden z důvodů
úspory místa.

> dC := diff(SQR,C1)=0:
> Vykon := (solve(su[1],C1)=collect(solve(dC,C1),[Te]))*S*a;
> Vykon := subs(Bs,tau=solve(su[2],tau),Vykon):

Vykon := P = S a
(
−Te − Ti Σ3 − Σ2 − Ti Σ5 + Σ1

Σ5 + N − 2 Σ3

)
Jak je vidět, za předpokladu znalosti součinu S a, tepelné kapacity kontejneru

K, počateční Ti, konečné Tinf a teploty okolí Te je možné pomoci proměřením
průběhu teploty povrchu kontejneru v čase určit zbytkový výkon paliva jako ana-
lytickou funkci naměřené teplotní závislosti.

Simulovaný ohřev

Předpokládejme, že sledujeme závislost teploty kontejneru naplněného vo-
dou, který zahřívá zbytkový výkon uloženého paliva. Základní vlastnosti kontej-
neru popisuje tabulka 1.

V případě výše uvedených hodnot je možné graficky znázornit průběh teploty
stěny kontejneru pro zbytkové výkony paliva od 5 kW do 30 kW s krokem 5 kW ,
viz obrázek 1.

Veličina Proměnná Velikost Jednotka
Hmotnost obalu Mo 75 000 kg

Hmotnost náplně Mo 5 000 kg

Měrné teplo obalu Co 520 J kg−1 K−1

Měrné teplo vody Cv 4186 J kg−1 K−1

Povrch kontejneru S 30 m2

Koeficient přestupu tepla alpha – α 25 W m−2 K−1

Počáteční teplota Ti 18 ◦C

Teplota okolí Te 25 ◦C

Tab. 1. Základní proměnné

> K := Mv*Cv+Mo*Co: SA := S*alpha: tau := K/SA: C1 := P/SA:
> Mo := 75000: Co := 520: Mv := 5000: Cv := 4186: S := 30:
> alpha := 25: Ti := 18: Te := 25:
> PS := [5000,10000,15000,20000,25000,30000]:
> TF := map(u->evalf(subs(P=u,limit(T,t=infinity))),PS):
> P1 := plot({seq(T,P=PS)}, t=0..5*86400,
> title="Predpokladany prubeh teploty",
> labels=["t[s]","T [C]"], color=black,thickness=3):
> P2 := plot(TF,t=0..5*86400,color=black,linestyle=4):
> P3 := plots[textplot]({seq([5*86400,TF[j]+1,
> cat(convert(PS[j]/1000,string)," kW")],j=1..6)},
> align={ABOVE,LEFT}):
> plots[display]({P1,P2,P3});

T∞ = 65.0

T∞ = 58.3

T∞ = 51.7

T∞ = 45.0

T∞ = 38.3

T∞ = 31.7

30 kW

25 kW

20 kW

15 kW

10 kW

5 kW

20

30

40

50

60

0 100 200 300 400

T [°C ]

t [s . 1000]

Obr. 1. Průběh teploty stěny kontejneru pro různé výkony

Zpracování simulovaného měření

Předpokládejme zbytkový výkon paliva v kontejneru 20 kW . Parametry kon-
tejneru uvažujme stejné jako v tabulce 1. Dále předpokládejme, že měření teploty
provádíme každou půlhodinu, časy měření jsou uloženy v seznamu tt a teploty
povrchu kontejneru měříme s přesností na 0.5 ◦C, teploty uložíme v seznamu Y.

> P := 20000; chyba := rand(-10..10)*0.05:
> tt := [seq(1800*j,j=0..240)]:
> Y := map(u->evalf(subs(t=u+1800*40,T)+chyba(),3),tt):
> P1 := plot(zip((u,v)->[u,v],tt,Y),symbol=circle,
> style=point,title="Simulovane teploty",
> color=black,symbolsize=15,
> labels=["cas [s]","t [◦C]"]):

Proměnná P1 obsahuje časový průběh simulovaných hodnot. Graf je součástí ob-
rázku 2.

Data proložíme obecnou exponenciální funkcí F = k1 e
−t
τ + K2. Funkce F

obsahuje neznámý nelineární parametr τ , neznámý lineární parametr k1 je v sou-
činu s nelineární funkcí τ . Tato skutečnost vylučuje možnost určit parametry
k1, τ, K2 funkce F pomocí lineární regrese – metody nejmenších čtverců. Použi-
jeme proto modifikovanou nelineární regresní iterační metodu odvozenou Gaus-
sem a Newtonem, upravenou Ganderem a Bartoňem [2, 3]. Z těchto parametrů
později určíme základní charakteristiku problému – zbytkový výkon paliva P

Prvním krokem je linearizace funkce podle proměnných k1 a τ v okolí počá-
tečních hodnot K1 a T.

> unassign(’Tau’,’tau’,’K1’,’T0’);
> F := exp(-t/tau)*k1+K2:
> e1 := mtaylor(F,[tau=Tau,k1=K1],2);

e1 := e−
t
T K1 + K2 + e−

t
T (k1 − K1) +

e−
t
T (τ − T) tK1

T2
.

Nyní nahradíme rozdíly k1 − K1 = DK, τ − T = DT opravami k počátečním
hodnotám.

> Dsu:=[k1-K1=DK,tau-Tau=DT]:
> e2:=subs(Dsu,e1);

e2 := e−
t
T K1 + K2 + e−

t
T DK +

e−
t
T DT tK1

T2
.

Proměnná e2 je již lineární vzhledm k opravám DK a DT pro počáteční hod-
noty K1 a T. Dále je samozřejmě lineární pro K2. Vytvoříme seznam koeficientů
u jednotlivých lineárních proměnných a dále vybereme ten člen, který lineární
koeficienty neobsahuje.

> Var := [DK,DT,K2];
> Col := map(u->select(has,e2,u)/u,Var);
> rest := remove(has,e2,Var);

7



Col :=
[
e−

t
T , e

− t
T t K1
T2 , 1

]
rest := e−

t
T K1

Nyní zbývá zvolit výchozí hodnoty K1 a T. Pokud je dosadíme do proměnné
Col a za čas t budeme postupně dosazovat číselné hodnoty ze seznamu tt dosta-
neme numerickou matici A. Pokud provedeme totéž s proměnnou rest získáme
seznam rozdílů naměřených hodnot Y a funkčních hodnot funkce F pro jednot-
livé časy ze seznamu tt. Při správné volbě výchozích hodnot K1 a T budou tyto
rozdíly malé. Opravy DK a DT pro výchozí hodnoty K1 a T a koeficient K2

minimalizují následující výraz, viz příkaz MAPLE leastsqrs [4].

min
[DK,DT,K2]

||A • [ DK, DT, K2 ] − [ rest ] ||2

Vypočtené opravy DK a DT pak přičteme k výchozím hodnotám K1 a T

a celý postup znovu zopakujeme. Výpočet – iteraci ukončíme v okamžiku, kdy
absolutní velikost opravy klesne pod stanovenou mez. V průběhu iterace zvý-
šíme numerickou přesnost na 24 platných cifer.

> K 1:=- 30: Tau := 75000:
> Digits := 24; epsilon := 10; n := 0;
> while epsilon > 10^(-Digits/2) do;
> n := n+1;
> A := matrix(evalf(map(u->subs(t=u,Col),tt))):
> B := evalf(zip((u,v)->u-subs(t=v,rest),Y,tt)):
> SOL := linalg[leastsqrs](A,B);
> epsilon:=max(map(u->abs(u),convert(SOL,list)[1..2])[]);
> if epsilon<1 then LM:=1 end if;
> K1:=K1+SOL[1]/LM; Tau:=Tau+SOL[2]/LM;
> print(krok=n,’epsilon’=epsilon);
> end do:

krok = 1, ε = 1586.57932207589427114091

krok = 2, ε = 1953.34519717514051488873

krok = 3, ε = 50.2249896781355783584539

krok = 4, ε = 0.747346442956547593010003

krok = 5, ε = 0.0109678001010463140302263

krok = 6, ε = 0.000160891760886171317598603

krok = 7, ε = 0.236018114804541723999455 10−5

krok = 8, ε = 0.346223726241167823815361 10−7

krok = 9, ε = 0.507888411927475432324407 10−9

krok = 10, ε = 0.745036077464870534422161 10−11

krok = 11, ε = 0.109364870794910660862430 10−12 .

Po ukončení iterace se vrátíme ke standardní numerické přesnosti, vyjádříme vý-
slednou funkci popisující průběh teploty v čase a vykreslíme ji spolu s naměře-
nýmy hodnotami, viz obrázek 2.

> Digits := 10;
> F := evalf(subs(tau=Tau,k1=K1,K2=SOL[3],F));
> Finf := limit(F,t=infinity);
> P2 := plot([F,Finf],t=0..max(tt[]*1.1),thickness=[5,1],
> linestyle=[1,3],color=red):
> plots[display]({P1,P2});

F := −13.72950375 e−0.00001272411817 t + 51.64362522

Finf := 51.64362522

38

52

0 100 200 300 400

T [°C ]

t [s.1000]

T∞ = 51.6

Obr. 2. Průběh teploty jako regresní funkce naměřených hodnot

Pokud nyní porovnáme vypočtené koeficienty s obecnými koeficienty z rov-
nice (5) je možné snadno dopočítat počáteční teplotu ti, výkon p a koeficient sa –
zde jsou na rozdíl od rovnice (5) označeny malými písmeny. Po dosazení k = K

a te = Te dostaneme konkrétní číselné hodnoty.

> e1 := Tau=k/sa: e2 := K1=ti-p/sa-te: e3 := SOL[3]=p/sa+te:
> Sol := solve({e3,e2,e1},{ti,p,sa});
> Sol := subs(k=K,te=Te,Sol);

Sol := { ti = 37.91412146, sa = 0.00001272411817 k,

p = 0.0006571195898 k − 0.00001272411817 te k }
Sol := { sa = 762.5564019, p = 20317.26697, ti = 37.91412146 } .

Porovnáme – li vypočtený výkon s předpokládanou hodnotou, zjistíme, že chyba
výsledku je přibližně 1.6%, což je velmi slušná shoda.

Nyní stanovíme korelační koeficient Korelacemezi naměřenými hodnotami
a funkčními hodnotami regresní funkce F, dále stanovíme průměrnou odchylku
pdy mezi naměřenou teplotou a funkční hodnotou funkce F a stanovíme střední
kvadratickou chybu sdy této odchylky.

> assign(Sol
> y := map(u->evalf(subs(t=u,F)),tt):
> Korelace:=stats[describe,linearcorrelation](y,Y);
> dy := zip((u,v)->u-v,y,Y):
> N := nops(tt):
> pd y:= sum(dy[i],i=1..N)/N;
> sdy := sqrt(sum((dy[i]-pdy)^2,i=1..N)/N);

Korelace := .9956710206
pdy := .3858921162 · 10−8

sdy := .3122672182

Vidíme, že průměrna odchylka pdy je zcela zanedbatelná a střední kvadra-
tická chyba sdy je přibližně dvě třetiny chyby, kterou byly zatíženy vstupní hod-
noty, viz Obrázek 3. Korelační koeficient blízký jedné vypovídá o vynikající shodě
regresní funkce s daty, viz Obrázek 4.

> plot([zip((u,v)->[u/1000,v],tt,dy),
> [[0,pdy],[tt[N]/1000,pdy]],[[0,pdy+sdy],
> [tt[N]/1000,pdy+sdy]],[[0,pdy-sdy],
> [tt[N]/1000,pdy-sdy]]],style=[point,line,line,line],
> thickness=[1,3,3,3],linestyle=[1,1,3,3],
> labels=["t [s]","DT [C]"],symbol=circle,
> color=[black,red,red,red],symbolsize=15,
> title="Znazorneni chyby vypoctu",
> labels=["t [s]","T [C]"]);
> plot(zip((u,v)->[u,v],Y,y),color=black,
> labels=["Y [C]","y [C]"],

title="Znazorneni kvality korelace");

Závěr

Z výše uvedených výpočtů vyplývá, že měřením teploty povrchu kontejneru
s vyhořelým jaderným palivem je možné určit zbytkový výkon paliva. Metodu
je však možné použít pokud známe tepelnou kapacitu naplněného kontejneru,
kterou je možné stanovit bud’ z technické dokumentace, nebo měřením.

Stanovení zbytkového výkonu z naměřeného průběhu teploty vyžaduje kon-
stantní teplotu a způsob proudění chladicího media, (voda, vzduch) v okolí kon-
tejneru. Tuto podmínky lze ale v prostředí úložiště vyhořelého paliva snadno
zajistit.

